


Parameter Estimation for Photovoltaic Diodes

▼ Introduction



The behavior of a photovoltaic diode is often modeled with an equivalent circuit (illustrated above), and is described by the equation below.

$$I_f = I_{pv} - I_0 \left(e^{\frac{I_f R_s + V_f}{n V_t}} - 1 \right) - \frac{I_f R_s + V_f}{R_p}$$

This application

- will rearrange this equation to give i in terms of the LambertW equation
- find the best-fit parameters against experimental data

References:

<http://www.hindawi.com/journals/jam/2013/362619/>

Gray, J.L., The Physics of the Solar Cell, in Handbook of Photovoltaic Science and Engineering, A. Luque, Hegedus, S., Editor. 2011, John Wiley and Sons

▼ Rearrange Diode Equation

> restart :

$$I_r := \text{solve} \left(I_f = I_{pv} - I_0 \cdot \left(e^{\frac{V_f + I_f \cdot R_s}{n \cdot V_t}} - 1 \right) - \frac{V_f + I_f \cdot R_s}{R_p}, I_f \right)$$

$$I_r := - \frac{ - \text{LambertW} \left(- \frac{ I_0 R_p R_s e^{\frac{R_p (I_0 R_s + I_{pv} R_s + V_f)}{n V_t (R_p + R_s)}}}{- R_p V_t n - R_s V_t n} \right) + \frac{ R_p (I_0 R_s + I_{pv} R_s + V_f)}{n V_t (R_p + R_s)} }{ R_s } n V_t + V_f$$

> $I_pred := \text{unapply}(I_r, V_f, I_{pv}, I_0, n, R_s, R_p) :$

▼ Import Experimental I-V Data for Photo Voltaic Diode

> data := ExcelTools:-Import("diode experimental data.xlsx", "Sheet1");

```
data := [ 1..26 x 1..2 Array  
         Data Type: anything  
         Storage: rectangular  
         Order: Fortran_order ]
```

> V_data := convert(data[.., 1], Vector) :

> I_data := convert(data[.., 2], Vector) :

> p1 := plot(V_data, I_data, style = point, legend = "Experimental Data") :

> T := 273.15 + 33 :

k := $1.380650 \cdot 10^{-23}$:

q := $1.602176 \cdot 10^{-19}$:

Vt := $\frac{k \cdot T}{q}$:

▼ Find Best-Fit Parameters

> res := Statistics:-NonlinearFit(I_pred, V_data, I_data, parameterranges = [0.1 ..1, 0 ..0.0001, 1 ..2, 0.01 ..0.1, 1 ..100], output = solutionmodule, iterationlimit = 50, optimalitytolerance = 0.01) :

> pars := res:-Results(parametervector);

```
pars := [ 0.766393737504875  
         0.00000936308114823049  
         1.92385643932305  
         0.0160026044081741  
         51.3874600644081 ]
```

> res:-Results(residualsumofsquares)

0.001785240007

▼ Plot Model Curve Against Experimental Data

> p2 := plot('I_pred'(Vf, pars[1], pars[2], pars[3], pars[4], pars[5]), Vf = min(V_data) ..max(V_data), color = black, legend = ["Model Curve"], axesfont = [Arial], legendstyle = [font = [Arial]], size = [800, 500], gridlines) :

> plots:-display(p1, p2, title = "Parameter Estimation for Photovoltaic Diode", titlefont = [Arial, 18])

Parameter Estimation for Photovoltaic Diode

